

Problema della produzione dei monitor

Una azienda produce monitor per PC in tre diversi stabilimenti. Il costo di produzione di ciascun monitor varia a causa della diversa efficienza produttiva degli stabilimenti. In particolare, la produzione di un monitor costa 90 Euro nello stabilimento di Rozzano, 105 Euro nello stabilimento di Prato e 100 Euro nello stabilimento di Frascati. La tabella seguente indica il costo in Euro da sostenere per spedire un monitor da ogni stabilimento ad ogni cliente.

	<i>IBM</i>	<i>HP</i>	<i>Toshiba</i>	<i>Olidata</i>
<i>Rozzano</i>	2.5	5.0	6.4	3.5
<i>Prato</i>	3	2.6	1.0	1.4
<i>Frascati</i>	1.4	3.4	2.0	5.5

I tre stabilimenti possono produrre, rispettivamente, 8 500, 9 200 e 11 000 monitor al mese. In base alle previsioni sulle vendite, la domanda mensile dei fabbricanti di PC è uguale a 5 000, 8 300, 6 300 e 2 700 monitor, rispettivamente. Per bilanciare la produzione si richiede che nell'impianto con costo maggiore (quello di Prato) siano prodotti almeno la metà dei monitor prodotti presso ciascuno degli altri impianti.

Si formuli un piano di produzione che minimizza i costi e si risolva il problema per mezzo di AMPL.

Formulazione

Parametri:

1. per $i = 1, \dots, 3$ sia c_i il costo di produzione di un monitor nello stabilimento i
2. per $i = 1, \dots, 3$, per $j = 1, \dots, 4$ sia s_{ij} il costo di spedizione di un monitor dallo stabilimento i al cliente j
3. per $i = 1, \dots, 3$ sia U_i il limite sulla produzione di monitor nello stabilimento i
4. per $j = 1, \dots, 4$ sia L_j il limite sulla richiesta di monitor del cliente j

Variabili decisionali:

per $i = 1, \dots, 3$, per $j = 1, \dots, 4$ sia x_{ij} il numero di monitor prodotti nello stabilimento i per il cliente j

Funzione obiettivo:

$$\min_{x_{ij}} \sum_{i=1}^3 \sum_{j=1}^4 (c_i + s_{ij}) x_{ij}$$

Vincoli:

1. per $i = 1, \dots, 3$: $\sum_{j=1}^4 x_{ij} \leq U_i$ (produzione)
2. per $j = 1, \dots, 4$: $\sum_{i=1}^3 x_{ij} \geq L_j$ (richiesta)
3. $\sum_{j=1}^4 x_{2j} - \frac{1}{2} \sum_{j=1}^4 x_{1j} \geq 0$ (Prato / Rozzano)
4. $\sum_{j=1}^4 x_{2j} - \frac{1}{2} \sum_{j=1}^4 x_{3j} \geq 0$ (Prato / Frascati)

Limiti:

per ogni $i = 1, \dots, 3$, $j = 1, \dots, 4$ si ha che $x_{ij} \in \mathbb{N}$

Modello e dati AMPL

- File azienda.mod:

```
# azienda.mod: problema dell'azienda produttrice di monitor

set STABILIMENTI;
set CLIENTI;

param costo_produzione { STABILIMENTI } >= 0;
param costo_spedizione { STABILIMENTI, CLIENTI } >= 0;
param limite_produzione { STABILIMENTI } >= 0;
param limite_richiesta { CLIENTI } >= 0;

var x { STABILIMENTI, CLIENTI } >= 0, integer;

minimize costo_totale: sum {i in STABILIMENTI, j in CLIENTI}
  ( costo_produzione[i] + costo_spedizione[i,j] ) * x[i,j];

subject to produzione {i in STABILIMENTI}:
  sum {j in CLIENTI} x[i,j] <= limite_produzione[i];

subject to richiesta {j in CLIENTI}:
  sum {i in STABILIMENTI} x[i,j] >= limite_richiesta[j];

subject to prato_rozzano:
  sum {j in CLIENTI} x["Prato",j] -
  (1/2) * sum {j in CLIENTI} x["Rozzano",j] >= 0;

subject to prato_frascati:
  sum {j in CLIENTI} x["Prato",j] -
  (1/2) * sum {j in CLIENTI} x["Frascati",j] >= 0;
```

- File azienda.dat:

```
# azienda.dat - con azienda.mod

set STABILIMENTI := Rozzano Prato Frascati ;
set CLIENTI := IBM HP Toshiba Olidata ;

param : costo_produzione limite_produzione :=
Rozzano 90 8500
Prato 105 9200
Frascati 100 11000;

param costo_spedizione :
IBM HP Toshiba Olidata :=
Rozzano 2.5 5.0 6.4 3.5
Prato 3 2.6 1.0 1.4
Frascati 1.4 3.4 2.0 5.5 ;

param limite_richiesta :=
IBM 5000
HP 8300
Toshiba 6300
Olidata 2700 ;
```

Variatione di un vincolo

Supponendo di rimuovere i vincoli che esprimono il fatto che lo stabilimento di Prato deve produrre almeno la metà dei monitor prodotti in ognuno degli altri stabilimenti, ed inserendo invece la richiesta che lo stabilimento di Prato produca almeno un terzo dei monitor prodotti congiuntamente negli altri due stabilimenti, si deve rimpiazzare i vincoli “Prato / Rozzano” e “Prato / Frascati” nella formulazione con il vincolo “Prato / altri”

$$\sum_{j=1}^4 x_{2j} - \frac{1}{3} \left(\sum_{j=1}^4 x_{1j} + \sum_{j=1}^4 x_{3j} \right) \geq 0.$$

Il file `azienda.mod` va cambiato: i vincoli `prato_rozzano` e `prato_frascati` vanno sostituiti con

```
subject to prato_altri:
  sum {j in CLIENTI} x["Prato",j] -
  (1/3) * ( sum {j in CLIENTI} x["Rozzano",j] +
           sum {j in CLIENTI} x["Frascati",j] ) >= 0;
```

Effettuando questa modifica, la soluzione del problema a variabili intere differisce da quella a variabili continue.